

Excel. Функции ИНДЕКС и ПОИСПОЗ

Это пятая глава книги Билла Джелена. [Всё о ВПР: от первого применения до экспертного уровня.](#)

Для извлечения данных вместо ВПР можно использовать связку функций ИНДЕКС и ПОИСКПОЗ. Формулы с ними работают быстрее и намного гибче. Лично я использую ВПР 90% времени, но иногда переключаюсь на ИНДЕКС + ПОИСКПОЗ, например, когда «ключевой» столбец не является первым слева в таблице.

Для меня функция ИНДЕКС совершенно непонятна. Проблема: я недавно прочитал справку Excel по этой функции. Кто в здравом уме будет использовать =ИНДЕКС(B4:G22;2;4), чтобы указать на ячейку F6 (рис. 5.1)?

The screenshot shows an Excel spreadsheet with a table of data. The formula bar at the top displays the formula =ИНДЕКС(B4:G22;2;4). A red arrow points from the formula bar to cell F6, which contains the value 7. A dialog box titled 'Аргументы функции' (Function Arguments) is open, showing the following arguments for the INDEX function:

Артикул	Месяц
A302	Апр Май Июнь
A397	114 90 1
A806	49 71 83 37 110 14
A936	2 101 18 63 54 95
A740	98 20 80 59 96 16
A945	36 19 106 52 35 70

Артикул: A397
Месяц: Апр

Какая строка? 2
Какая колонка? 4
Результат? 7

Аргументы функции

ИНДЕКС

Массив: B4:G22 = {99;41;31;114;90;1:10;56;3;7;44;50;49...}

Номер_строки: 2 = 2

Номер_столбца: 4 = 4

= 7

Возвращает значение или ссылку на ячейку на пересечении конкретных строки и столбца, в данном диапазоне.

Массив: диапазон ячеек или константа массива.

Значение: 7

[Справка по этой функции](#) [ОК] [Отмена]

Рис. 5.1. Функция ИНДЕКС кажется бесполезной

Стратегия: вы никогда не будете использовать ИНДЕКС, не используя ПОИСКПОЗ в качестве второго или третьего аргумента. Поэтому давайте вернемся к функции ИНДЕКС после того как вы увидите, что делает ПОИСКПОЗ.

На самом деле, вы уже знакомы с тем, как работает функция ПОИСКПОЗ. Действительно, если вы используете ВПР, вы уже знаете кое-что и о ПОИСКПОЗ. Давайте сравним эти функции (рис. 5.2):

- Первым параметром является поисковое значение, как ВПР.
- Таблица подстановки включает один столбец, а не прямоугольный диапазон.
- Вам не нужно указывать номер столбца, поэтому отсутствует третий аргумент.
- Последний аргумент – ЛОЖЬ, как и в ВПР (также можно использовать ноль).

Пока всё очень похоже.

А вот и отличие... ПОИСКПОЗ не возвращает значение из таблицы. ПОИСКПОЗ расскажет вам, какая строка в таблице содержит искомое значение. Какая... по номеру. В нашем примере – вторая.

Трюк заключается в том, что вы всегда будете использовать ПОИСКПОЗ внутри функции ИНДЕКС. Итак, вернемся к индексу.

СУММ : X ✓ fx =ПОИСКПОЗ("A397";SA\$4:SA\$22;ЛОЖЬ)

	A	B	C	D	E	F	G	H	I	J	K
1		2									
2											
3	Артикул	Янв	Фев	Мар	Апр	Май	Июн				
4	A302	99	41	31	114	90	1				
5	A397	10	56	3	7	44	50				
6	A806	49	71	83	37	110	14				
7	A936	2	101	18	63	54	95				
8	A740	98	20	80	59	96	16				
9	A945	36	19	106	52	35	70				
10	A695	62	43	58	34	51	69				
11	A507	105	91	78	29	32	75				
12	A571	13	108	27	67	86	68				
13	A315	11	9	4	76	30	23				
14	A552	55	26	28	85	111	81				
15	A628	40	87	88	47	107	60				
16	A621	74	57	103	33	104	64				
17	A211	93	92	84	61	112	79				
18	A563	53	15	8	42	24	46				
19	A547	94	12	77	38	22	39				
20	A940	48	100	102	97	66	45				
21	A339	25	72	73	113	21	6				
22	A673	17	89	82	65	5	109				
23											

Рис. 5.2. ПОИСКПОЗ – замаскированная ВПР

Стратегия: фокус в том, чтобы использовать ИНДЕКС для возврата значения. Примените при этом ПОИСПОЗ в качестве второго и/или третьего аргумента, чтобы налету вычислить к какой строке или столбцу обратиться. (В примере выше функция ПОИСКПОЗ искала в вертикальном списке. Она также можете найти значение в горизонтальном списке. По сути, аналог ГПР.)

Далее будут приведены примеры использования связки ИНДЕКС + ПОИСКПОЗ для решения проблем.

Поиск слева от ключевой колонки. Проблема: таблица подстановки ведется в другом отделе. Колонка с ценой у них расположена слева от столбца с номенклатурой. Могу ли я указать «-1» (минус один) в качестве третьего аргумента функции ВПР?

	F	G	H	I
1	Цена	Артикул	Описание	
2	37,95	BG33-3	14K Gold Bangle Brace	
3	27,95	CR50-3	14K Gold Cross with O	
4	54,95	RG75-3	14K Gold RAY OF LIGHT	
5	39,95	RG78-25	14K Gold Ballerina Rin	
6	45,95	W25-6	18K Italian Gold Wom	
7	28,95	BR26-3	18K Italian Gold Men's	
8	53,95	BR15-3	14K Gold Onyx Men's I	
9	29,95	BG33-8	14K Gold Bangle Brace	
10	34,95	BG33-17	14K Gold Bangle Brace	
11	31,95	CR 50-4	14K Gold Onyx Cross	
12	55,95	CR50-2	14K Gold Onyx Cross	
13	55,95	CR50-1	14K Gold Onyx Cross	
14	56,95	Cross50-5	14K Gold Onyx Cross v	
15	46,95	CR50-6	14K Gold Onyx Cross v	
16	44,95	ER41-4	14K Gold Swiss Cut Eai	

Рис. 5.3. Поиск значение слева от номенклатуры

Стратегия: к сожалению, команда разработчиков Excel не предоставляет возможности для ВПР искать слева от ключевого поля. Иногда решение может заключаться в том, чтобы просто вырезать столбец F и вставить его справа от столбца с артикулами, например, в столбце I. И после этого применить ВПР. Мы исходим из того, что такие манипуляции вам недоступны. Вы можете использовать ПОИСКПОЗ, чтобы выяснить, какую цену извлечь из таблицы для того или иного артикула.

Итак, артикулы расположены в столбце G. Они могут быть не отсортированы, но не должно быть дублей. Посмотрите на формулу в С6 (рис. 5.4) =ПОИСКПОЗ(A6;\$G\$2:\$G\$29;0). Она говорит Excel найти CR50-3 в диапазон G2:G29. Третий аргумент – ноль – указывает на то, что вы ищете точное совпадение.

СУММ : X ✓ fx =ПОИСКПОЗ(A6;\$G\$2:\$G\$29;0)

	A	B	C	E	F	G	H	I
1	Артикул	Количес	Цена		Цена	Артикул	Описание	
2	ER46-22	30	51,95		37,95	BG33-3	14K Gold Bangle Brac	
3	ER41-4	18	44,95		27,95	CR50-3	14K Gold Cross with	
4	BR15-3	18	53,95		54,95	RG75-3	14K Gold RAY OF LIGI	
5	CR50-1	12	55,95		39,95	RG78-25	14K Gold Ballerina Ri	
6	CR50-3	18	2		45,95	W25-6	18K Italian Gold Won	
7					28,95	BR26-3	18K Italian Gold Men	
8					53,95	BR15-3	14K Gold Onyx Men's	
9					29,95	BG33-8	14K Gold Bangle Brac	

Рис. 5.4. ПОИСПОЗ находит CR50-3 во второй строке таблицы подстановки

ПОИСКПОЗ возвращает значение 2, т.е. говорит, что CR50-3 находится во второй строке диапазона. И на самом деле CR50-3 находится в G3. ПОИСКПОЗ возвращает относительное положение элемента в пределах диапазона поиска. Ответ «2» говорит, что CR50-3 находится во второй ячейке G2:G29.

Теперь, когда вы знаете позицию элемента в таблице поиска, вы можете использовать функцию ИНДЕКС, чтобы извлечь цену из таблицы. Укажите диапазон цен в качестве первого аргумента функции ИНДЕКС (рис. 5.5). Второй аргумент задает строку в таблице подстановки. И этим вторым аргументом является функции ПОИСКПОЗ. Поскольку у вас только один столбец в таблице подстановки, вам не нужен третий аргумент функции ИНДЕКС. В ячейке С6 формула =ИНДЕКС(\$F\$2:\$F\$29;ПОИСКПОЗ(A6;\$G\$2:\$G\$29;0)).

СУММ : X ✓ fx =ИНДЕКС(\$F\$2:\$F\$29;ПОИСКПОЗ(A6;\$G\$2:\$G\$29;0))

	A	B	C	D	E	F	G	H	I	J
1	Артикул	Количес	Цена	Итого		Цена	Артикул	Описание		
2	ER46-22	30	51,95	\$1 558,50		37,95	BG33-3	14K Gold Bangle Bracelet witi		
3	ER41-4	18	44,95	\$809,10		27,95	CR50-3	14K Gold Cross with Onyx		
4	BR15-3	18	53,95	\$971,10		54,95	RG75-3	14K Gold RAY OF LIGHT Onyx		
5	CR50-1	12	55,95	\$671,40		39,95	RG78-25	14K Gold Ballerina Ring w/ Bl		
6	CR50-3	18	27,95	\$503,10		45,95	W25-6	18K Italian Gold Women's Wa		
7	Итого	96		\$4 513,20		28,95	BR26-3	18K Italian Gold Men's Bracel		
8						53,95	BR15-3	14K Gold Onyx Men's Bracele		
9						29,95	BG33-8	14K Gold Bangle Bracelet witi		

Рис. 5.5. По существу, вы применили «левую» ВПР

Быстрый поиск по множеству столбцов. Проблема: мне нужно сделать двенадцать столбцов поиска. Таблица подстановки очень большая. Набор данных также весьма велик. Значения постоянно меняются и формулы пересчитываются.

ВПР – это дорогая (по ресурсам) функция. Довольно много времени нужно, чтобы найти точное соответствие в таблице подстановки. Возможно придется искать среди 200 строк таблицы подстановки, чтобы найти артикул при поиске январского значения. Когда Excel начнет искать февральское значение, он должен всё начать заново. По времени, ПОИСКПОЗ и ВПР работают приблизительно одинаково. Зато ИНДЕКС работает существенно быстрее, напрямую обращаясь к указанной строке и считывая значение.

Стратегия: добавить столбец «Где?», который вскоре будет скрыт. Использовать в нем формулу ПОИСКПОЗ, возвращающую номер строки в таблице подстановки, где находится артикул (рис. 5.6).

=ПОИСКПОЗ(A3;\$Q\$3:\$Q\$226;0)

	A	B	C	D	E	F
1						
2	Арти	Где	Янв	Фев	Мар	Апр
3	A308	208				
4	A219	119				
5	A249	149				
6	A154	54				
7	A128	28				

Рис. 5.6. Продукт A308 находится в 208-й строке таблицы подстановки

Теперь можно использовать функцию ИНДЕКС в 12 столбцах основной таблицы для извлечения значений из соответствующего столбца таблицы подстановки (рис. 5.7). Важно спланировать размещение знаков доллара в формуле:

- В ячейку C3 введите =ИНДЕКС(R\$3:R\$226;\$B3;0). Вы используете \$ перед 3 и 226, чтобы ссылка не менялась при копировании формулы вдоль столбца C. Однако, вы не используете знаки доллара перед буквой R. При копировании этой части формулы вправо в ячейку D3 ссылка изменится на S\$3:S\$226. Будет использован столбец S таблицы подстановки, относящийся к февралю. Второй аргумент функции ИНДЕКС использует один знак доллара перед обозначением столбца B. При копировании формулы вправо она по-прежнему будет ссылаться на столбец B.
- Скопируйте C3 в диапазон D3:H3.
- Выберите C3:H3. Скопируйте этот диапазон вниз на все строки, дважды щелкнув маркер заполнения в правом нижнем углу ячейки H3.
- Скройте столбец B.

ВПР : X ✓ fx =ИНДЕКС(V\$3:V\$226;\$B8;0)

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AAA	AAE	AC
1																													
2	Арти	Где?	Янв	Фев	Мар	Апр	Май	Июн	Июл	Авг	Сен	Окт	Ноя	Дек			Арти	вз	ар	тр	ай	нл	яг	ен	кт	оя	ек		
3	A308	208	6	1	9	2	9	1	5	4	4	4	7	3			A101	5	2	0	8	3	1	1	0	6	1	4	7
4	A219	119	0	0	5	0	2	8	8	7	8	3	0	0			A102	3	0	7	0	4	9	4	3	5	3	7	1
5	A249	149	1	7	1	3	2	9	6	7	1	2	9	5			A103	1	4	9	0	3	4	3	3	8	0	5	6
6	A154	54	2	5	8	0	3	3	5	5	2	8	9	3			A104	0	4	2	5	0	0	9	7	6	8	8	5
7	A128	28	4	1	8	6	2	5	7	8	4	0	9	5			A105	9	4	8	1	3	4	5	7	7	7	9	7
8	A229	129	5	5	4	7	8;0)	1	5	6	9	2	7	3			A106	6	3	6	2	6	2	1	1	3	8	7	5
9	A111	11	0	7	1	1	5	8	2	5	2	2	6	2			A107	7	6	4	0	9	4	7	4	3	7	3	1
10	A235	135	7	6	1	3	1	7	6	0	4	9	7	5			A108	3	9	3	8	6	8	0	4	9	7	1	7

Рис. 5.7. Такая таблица в десять раз быстрее, чем аналогичная, использующая функции ВПР

Решение на основе ПОИСКПОЗ + ИНДЕКС, показанное здесь, позволяет решить проблему редактирования третьего аргумента функции ВПР описанного [ранее](#).

Использование функции ПОИСКПОЗ для нахождения чуть большего значения. Проблема: мне нужно найти продукт, объем которого равен или больше, чем искомое значение. Я продаю порошок для покрытия, расфасованный в контейнеры различного размера. Содержимое каждого контейнера может покрыть определенную площадь (в квадратных метрах). Если клиенту необходимо покрыть N квадратных метров, мне нужно найти ближайший (но больший) контейнер.

ВПР позволяет найти следующий меньшего размера, но не следующий большего размера (см. главу 6). На рис. 5.8 показан «правильный» продукт для покрытия 131 кв. м. Это продукт с артикулом N2557, который позволяет покрыть до 144 кв. м.

ВПР : X ✓ fx =ИНДЕКС(\$I\$2:\$I\$9;ПОИСКПОЗ(B2;\$H\$2:\$H\$9;-1))

	A	B	C	D	EFG	H	I	J	K
	Заказ	Требуется покрыть, кв. м	Рекомендованный продукт	Стоимость		Покрывает, кв. м	Продукт	Цена за единицу	Общая стоимость
1									
2	1001	131	N2557	\$1 144,80		512	N2558	\$6,95	\$3 558,40
3	1002	445	N2558	\$3 558,40		144	N2557	\$7,95	\$1 144,80
4	1003	144	N2557	\$1 144,80		96	N2556	\$8,25	\$792,00
5	1004	10	N2553	\$107,40		48	N2555	\$8,45	\$405,60
6	1005	1	N2551	\$9,95		24	N2554	\$8,75	\$210,00
7	1006	160	N2558	\$3 558,40		12	N2553	\$8,95	\$107,40
8	1007	1	N2551	\$9,95		6	N2552	\$9,25	\$55,50
9	1008	57	N2556	\$792,00		1	N2551	\$9,95	\$9,95
10	1009	99	N2557	\$1 144,80					

Рис. 5.8. Используя в качестве последнего аргумента функции ПОИСКПОЗ «-1» можно найти следующее большее значение

Стратегия: используйте ПОИСКПОЗ с третьим аргументом равным -1 (минус 1). Если использование в качестве третьего аргумента нуля, позволяет найти точное совпадение, то при «-1» ПОИСКПОЗ ищет значение, равное или ближайшее большее, чем искомое значение. Например, ПОИСКПОЗ(B2;\$H\$2:\$H\$9;-1) вернет 2, поскольку 144 кв. м – ближайшее большее искомых 131 кв. м. A =ИНДЕКС(\$I\$2:\$I\$9;2;-1) вернет N2557.

Обратите внимание, что для использования функции ПОИСКПОЗ с аргументом «-1», таблица подстановки должна быть отсортирована в порядке убывания. Также заметьте, что ПОИСКПОЗ может найти значение *равное* искомому. Например, в строке 4 поиск по значению 144 вернул артикул N2557, а не следующий большего размера.

Чтобы закончить анализ, добавьте в ячейку D2 формулу =ИНДЕКС(\$K\$2:\$K\$9;ПОИСКПОЗ(B2;\$H\$2:\$H\$9;-1)) или =ВПР(C2;\$I\$2:\$I\$9;3;ЛОЖЬ).

Использование ПОИСКПОЗ для определения, присутствует ли элемент в списке. Проблема: у меня есть список элементов (товаров), по которым сегодня поступил заказ. Мне нужно сравнить этот список со списком позиций, которые есть в наличии на складе. Это же вопрос был задан в начале [главы 4](#). Однако, сейчас с ПОИСКПОЗ ответ может быть немного проще.

	A	B	C	D
1	Товары в заказе	В наличии?		
2	BR15-3			Товары на складе
3	ER49-20			BG33-17
4	BG33-17			BG33-3
5	BR15-3			BR15-3
6	CR50-6			CR50-2
7				Cross50-5
8				ER 46-33
9				ER41-4
10				ER46-28
11				ER46-7
12				ER49-21
13				ER80-63
14				P411A
15				RG75-3
16				W25-6
17				

Рис. 5.9. Какие элементы из столбца С есть в наличии, т.е. присутствуют в столбце D?

Стратегия: введите в B2 формулу =ПОИСКПОЗ(A2;\$D\$3:\$D\$16;0). Если элемент найден в списке *Товары на складе*, ПОИСКПОЗ вернет число – номер элемента в списке. Если товара нет в наличии, ПОИСКПОЗ вернет ошибку #Н/Д. Воспользовавшись рекомендациями главы 4, можно с помощью функций ЕСЛИ и ЕОШИБКА модифицировать формулу, чтобы она возвращала что-то более человеческое, например, =ЕСЛИ(ЕОШИБКА(ПОИСКПОЗ(A2;\$D\$3:\$D\$16;0));"нет на складе";"в наличии").

=ЕСЛИ(ЕОШИБКА(ПОИСКПОЗ(A2;\$D\$3:\$D\$16;0));"нет на складе";"в наличии")					
	A	B	C	D	E
1	Товары в заказе	В наличии?			
2	BR15-3	в наличии		Товары на складе	
3	ER49-20	нет на складе		BG33-17	
4	BG33-17	в наличии		BG33-3	
5	BR15-3	в наличии		BR15-3	
6	CR50-6	нет на складе		CR50-2	
7				Cross50-5	
8				ER 46-33	
9				ER41-4	
10				ER46-28	
11				ER46-7	
12				ER49-21	
13				ER80-63	
14				P411A	
15				RG75-3	
16				W25-6	
17					

Рис. 5.10. Использование ПОИСКПОЗ для определения, присутствует ли элемент в списке